# $Id: PKGBUILD 76743 2012-09-28 19:43:00Z spupykin $
# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>

pkgname=vile
pkgver=9.8_i
_basever=9.8
pkgrel=1
pkgdesc="vi like emacs"
arch=('i686' 'x86_64')
url="http://invisible-island.net/vile/vile.html"
depends=('ncurses')
license=('custom')
options=(!libtool)
source=("ftp://invisible-island.net/vile/vile-${_basever}.tgz"
	"ftp://invisible-island.net/vile/patches/vile-${_basever}a.patch.gz"
	"ftp://invisible-island.net/vile/patches/vile-${_basever}b.patch.gz"
	"ftp://invisible-island.net/vile/patches/vile-${_basever}c.patch.gz"
	"ftp://invisible-island.net/vile/patches/vile-${_basever}d.patch.gz"
	"ftp://invisible-island.net/vile/patches/vile-${_basever}e.patch.gz"
	"ftp://invisible-island.net/vile/patches/vile-${_basever}f.patch.gz"
	"ftp://invisible-island.net/vile/patches/vile-${_basever}g.patch.gz"
	"ftp://invisible-island.net/vile/patches/vile-${_basever}h.patch.gz"
	"ftp://invisible-island.net/vile/patches/vile-${_basever}i.patch.gz")
md5sums=('b5a0d89165f633a662cdb4b5c57f2e2f'
         '685cb681943e9315a1689a6a6f4734cb'
         'f3842ea427c635d0ee96f6da8566fa56'
         '69cfdbb0b84a51802455bbe44a5ce32e'
         '5d675193bc02ac4cab72452bf0051489'
         '1e6317a15d7b2cd99f921f3af312b977'
         '7f8a38272395f07e1377cc1943d37e7f'
         '26ab76a7305b7ec97b22855810911ceb'
         '6ea4cf22e29f5418f26ba6632985c70e'
         '908e8f9cf3c9a681c3ef76a3cd04a0c3')

build() {
  cd $srcdir/$pkgname-${_basever}

  for i in $srcdir/vile-${_basever}?.patch; do
    patch -p1 <$i
  done
  sed -i 's|FLEX_BETA|FLEX_NOBETA|g' filters/filters.h

  ./configure --prefix=/usr
  make
}

package() {
  cd $srcdir/$pkgname-${_basever}
  make install DESTDIR=$pkgdir

  mkdir -p $pkgdir/usr/share/licenses/$pkgname
  cat <<EOF >$pkgdir/usr/share/licenses/$pkgname/license
Copyright

Vile is copyrighted by Paul G. Fox, Thomas E. Dickey and Kevin Buettner with some
files (e.g., visvile) copyright by Clark Morgan.
We distribute it under the terms of the GNU Public License, Version 2.
EOF
}
